
Abstract. In this paper we describe an Agda-based formalization of
results from Üresin & Dubois’ “Parallel Asynchronous Algorithms for
Discrete Data.” That paper investigates a large class of iterative algo-
rithms that can be transformed into asynchronous processes. In their
model each node asynchronously performs partial computations and
communicates results to other nodes using unreliable channels. Üresin
& Dubois provide sufficient conditions on iterative algorithms that guar-
antee convergence to unique fixed points for the associated asynchronous
iterations. Proving such sufficient conditions for an iterative algorithm
is often dramatically simpler than reasoning directly about an asyn-
chronous implementation. These results are used extensively in the liter-
ature of distributed computation, making formal verification worthwhile.

Our Agda library provides users with a collection of sufficient condi-
tions, some of which mildly relax assumptions made in the original paper.
Our primary application has been in reasoning about the correctness of
network routing protocols. To do so we have derived a new sufficient con-
dition based on the ultrametric theory of Alexander Gurney. This was
needed to model the complex policy-rich routing protocol that maintains
global connectivity in the internet. Additionally we highlight and discuss
two propositions from Üresin & Dubois, which during the course of the
formalisation, turned out to be false.

1 Introduction

Many applications work with an iterative algorithm F and an initial state x(0)
where successive states are computed as

x(t + 1) = F(x(t))

until a fixed point ξ is reached at some time t′ when x(t′) = ξ = F(ξ). Here we
assume that x(t) represents an n-dimensional vector in some state space. If we
rewrite F as

F(x) = (F1(x), . . . , Fn(x)),

then we can imagine that it may be possible to assign the computation of each
Fi to a distinct processor. This might be performed in parallel with shared
memory or in a completely distributed manner. However, enforcing correctness
using global synchronization mechanisms may incur performance penalties that
negate the gains from the parallelization. Furthermore, global synchronization
is infeasible for applications such as network routing.

This leads to the question: When can we use the Fi to correctly implement
an asynchronous version of F-iteration? There are many answers to this question
that depend on properties of the state space and the function F – see the survey
paper by Frommer & Syzld [9].

Many of the approaches discussed in [9] rely on the rich structure of vector
spaces over continuous domains. However, our motivation arises from network
routing protocols where the state space is comprised of discrete data. Happily,
Üresin and Dubois [21] have developed a theory of asynchronous iterations over
discrete state spaces. They prove that if F is an Asynchronously Contracting
Operator (ACO, see Sect. 3), then the associated asynchronous iteration will
always converge to the correct fixed point. Their proof uses very weak assump-
tions about inter-process communication (indeed, in the case that the state
space is finite they show that ACO is a necessary condition as well). These
weak assumptions are a good model for the case of distributed routing protocols
where messages can be delayed, lost, duplicated or reordered. Henceforth we will
refer to Üresin and Dubois [21] as UD.

Proving that a given F is an ACO can be dramatically simpler than reasoning
directly about an asynchronous implementation. However, in many cases it still
remains non-trivial and so UD also derive several sufficient conditions that imply
the ACO condition. These conditions are typically easier to prove for many
common iterative algorithms. For example, they provide sufficient conditions for
special cases where the state space is a partial order and F is order preserving.

In this paper we describe an Agda [3] formalization of the sufficient conditions
and associated proofs from UD. This represents one part of a larger project
in which we are developing formalized proofs of the asynchronous convergence
for policy-rich distributed Bellman-Ford routing protocols (see [5]). This work
required formalizing a new sufficient condition not found in UD, based on the
ultrametric theory of Gurney [11]. During formalization it also became apparent
that two of the other sufficient conditions in the original paper are incorrect. We
provide a counter-example. In addition we suggest how to strengthen one of the
sufficient conditions so that correctness is still guaranteed.

Many other applications of the results of UD can be found in the literature
(for example, [4,6,7,16]). The proofs in UD are mathematically rigorous in the
traditional sense, but their definitions are somewhat informal and they occasion-
ally claim the existence of objects without providing an explicit construction. In
our opinion a formal verification of the results is therefore a useful exercise.

There have been other efforts to formalize asynchronous computation such
as Meseguer and Ölveczky [17] for real-time systems and Henrio, Khan, and
Kammüller [13,14] for distributed languages. However, as far as we know our
work is the first attempt to formalize the results of UD.

Our Agda development can be found on Github [1]. We hope that this will
be a valuable resource for others interested in asynchronous iterations.

2 Preliminaries

In this section we introduce the components of the model of asynchronous com-
putation that underpin UD’s results together with their Agda formalizations.
Naturally, when formalizing mathematical proofs, there are concerns over steps
that are considered trivial in the informal proof. We therefore highlight key fea-
tures in the proof which are in practice significantly more complex than perhaps
implied by the original reasoning.

Definition 1. An iterative algorithm consists of an initial state x(0) and an
operator F such that ∀t ∈ N, x(t + 1) = F(x(t)).

We begin by formalizing the product state space S = S1 × · · · × Sn. This is
encoded by a Fin n-indexed family of Setoids. The type S is a function that
takes i and returns the Carrier type of the i-th setoid. We can now formalize the
iterative algorithm as follows:

sync-iter : S → N → S
sync-iter x0 zero = x0
sync-iter x0 (suc K) = F (sync-iter x0 K)

Routing Example. We briefly outline how this work can be applied to reason-
ing about convergence of a very general class of internet routing protocols. Full
details can be found in Daggitt et al. [5].

Routing problems can be formalized as a tuple (R, ⊕, E, 0̄, ∞), where:

– R is the set of routes,
– ⊕ : R → R → R is the choice operator, returning the preferred route,
– E is a set of functions of the form R → R representing generalized edge

weights,
– 0̄ is the trivial route from a node to itself,
– ∞ is the invalid route.

A network configuration is represented as an n × n adjacency matrix A over
E. The state space is made up of n × n matrices X over R. Matrix addition,
X ⊕ X′, is just the pointwise application of ⊕. The application of A to state X
is defined as

(A(X))ij =

(⊕
k

Aik(Xkj)

)
.

That is, each node i choose the best extensions of the routes to j advertised by
its neighbors. Finally, the iterative algorithm F is defined as

F(X) = A(X) ⊕ I, (1)



626 R. Zmigrod et al.

where I is the matrix defined as Iii = 0̄, and Iij = ∞ for i �= j. As explained in [5],
an asynchronous version of F provides a good model of Distributed Bellman-
Ford (DBF) routing protocols. At each asynchronous iteration in the distributed
setting, each node i will compute only the i-th row of F(X) from the rows
communicated by its adjacent neighbors.

Shortest paths routing is probably the simplest example where ⊕ = min and
E is the set of all fw with fw(r) = w + r.

2.1 Schedules

Schedules determine the asynchronous behaviour; they dictate when nodes
release new information and the timing of that information propagating to other
nodes. Let I be the set of nodes participating in the asynchronous process.

Definition 2. A schedule ζ is a pair of functions α : N → P(I) and β : N →
I → I → N which satisfy the following properties:

A1 : ∀t ∈ N, i, j ∈ I. β(t + 1, i, j) ≤ t
A2 : ∀t ∈ N, i ∈ I. ∃t′. t < t′ ∧ i ∈ α(t′)
A3 : ∀t ∈ N, i, j ∈ I. ∃t′. ∀t′′. t′ < t′′ ⇒ β(t′′, i, j) �= t

The activation function α takes a time t and returns a subset of I containing
the nodes that updated their value at time t. The data flow function β takes a
time t and two nodes i and j and returns the time at which the data used by i
at time t was generated by j.

Assumption A1 captures the notion of causality by ensuring that data can
only be used after it was generated. A2 says that each node continues to activate
indefinitely. Lastly, A3 says that the data generated at time t will only be used
for a finite number of future updates.

Generalization 1. UD use a shared-memory model with all nodes communi-
cating via shared memory, and so their definition of β takes only a single node
i. However this model does not capture processes in which nodes communicate
in a pairwise fashion without shared memory (e.g. internet routing). We have
therefore augmented our definition of β to take two nodes, a source and desti-
nation. Their original definition can be recovered by providing a β function that
is constant in its third argument.

Generalization 2. UD assumed that all nodes are active initially (i.e. α(0) =
I), which is unlikely to be true in a distributed context. Fortunately this assump-
tion turns out to be unnecessary.

We formalize schedules in Agda as a dependent record. The number of
nodes in the computation is passed as a parameter and the nodes them-
selves are represented by the Fin n type. The three properties are named
causality, nonstarvation, and finite respectively.

record Schedule (n : N) : Set where
field
α : (t : T) → Subset n
B : (t : T)(i j : Fin n) → T

causality : ∀ t i j → B (suc t) i j ≤ t
nonstarvation : ∀ t i → ∃ ń k → i ∈ α (t + suc k)
finite : ∀ t i j → ∃ ń k → ∀ l → B (k + l) i j �≡ t

In the definition we use T as an alias for N to help semantically differentiate
between times and other natural numbers. It would also be possible to implicitly
capture causality by changing the return type of B to Fin t instead of T. However,
it turns out that in practice when using B we nearly always want a regular time,
and therefore each call to B would require a conversion to T. We thus decide to
keep causality as an explicit field of Schedule.

Another choice made when designing the formalisation of nonstarvation and
finite was to replace the conditions such as ∀y. x ≤ y =⇒ P (y) with ∀y. P (x+y).
This removes the need to pass around proof terms, and consequently often makes
using these properties easier to use. This same technique is used throughout the
rest of our library.

An asynchronously iteration can be constructed by combining an iterative
algorithm with a schedule.

Definition 3. An asynchronous iteration over a schedule S = (α, β), an initial
state x(0), and an operator F, is denoted as (F, x(0), S ) such that ∀t ∈ N, i ∈ I

xi(t + 1) =

{
xi(t) if i /∈ α(t + 1)
Fi(x0(β(t + 1, i, 0)), . . . ,xn−1(β(t + 1, i, n − 1))) otherwise

We formalize this in Agda as follows:

async-Iter’ : Schedule n → S → ∀ {t} → Acc _<_ t → S
async-Iter’ S x0 {zero} _ i = x0 i
async-Iter’ S x0 {suc t} (acc rec) i with i ∈? α S (suc t)
... | yes _ = F (ń j → async-Iter’ S x0

(rec (B S (suc t) i j) (s≤s (causality S t i j))) j) i
... | no _ = async-Iter’ S x[0] (rec t ≤-refl) i

Those unfamiliar with Agda may wonder why the Acc argument is necessary.
While we can see that this function will terminate as each recursive call goes
from time t to time β(t, i, j) which is strictly smaller due to causality, the Agda
termination checker cannot detect this without help. Acc is a data-type found
in the Agda standard library that helps the termination checker by providing
an argument to the function that always becomes structurally smaller with each
recursive call. Using the proof that the natural numbers are well-founded, this
complexity is hidden from the user in the main function:

async-iter : Schedule n → S → T → S
async-iter S x0 t = async-iter’ S x0 (<-wellFounded t)

3 Convergence Theorem

UD define a class of Fs called Asynchronously Contracting Operators (ACOs).
They then prove that if an operator is an ACO, then it will converge to the
correct fixed point for all possible schedules.

Definition 4. An operator F is an asynchronously contracting operator (ACO)
on a subset D(0) of the state space S = S0 × S1 × · · · × Sn−1 iff there exists a
sequence of sets D(K) such that

(i) ∀K ∈ N. D(K) = D0(K) × D1(K) × · · · × Dn−1(K)
(ii) ∃ξ ∈ S. ∃T ∈ N. ∀K ∈ N.

K < T ⇒ D(K + 1) ⊆ D(K)
K ≥ T ⇒ D(K) = {ξ}

(iii) ∀K ∈ N. x ∈ D(K) ⇒ F(x) ∈ D(K + 1)

The sequence D(K) can be seen as a form of approximation for the process with
each iteration providing a higher accuracy. Each set contains the possible states
at a moment in time. D(0) contains many possible states as the algorithm has
just begun, and each set in the sequence removes some incorrect states. This
occurs until D(T ) = {ξ} when the converged state has been found.

Generalization 3. The definition of ACO in UD used the clause K < T ⇒
D(K +1) ⊂ D(K), where we have relaxed this to K < T ⇒ D(K +1) ⊆ D(K).
This relaxation is also found in the survey by Frommer and Szyld [9].

The definition of an ACO is captured in the following record type:

record ACO p : Set _ where
field
D : N → ∀ i → Si i → Set p
D-decreasing : ∀ K → D (suc K) ⊆ D K
D-finish : ∃2 ń T ξ → ∀ K → IsSingleton ξ (D (T + K))
F-monotonic : ∀ K {t} → t ∈ D K → F t ∈ D (suc K)

The variable p represents the universe level of the family of sets D, while the
universe level of ACO is inferred automatically (Set _). The sets themselves are
implemented as a double-indexed family of predicates over Si i.

The following theorem is the main sufficient condition proved in UD.

Theorem 1. If F is an ACO on a set D(0), then for all schedules S , any
asynchronous iteration x(k) = (F,x(0),S ) with x(0) ∈ D(0), converges to the
unique fixed point ξ of F in D(0).

In order to prove this theorem, UD consider the concept of a pseudo-periodic
schedule. It is then proved that every schedule (Definition 2) is in fact pseudo-
periodic, which greatly simplifies reasoning about schedules. This is perhaps the
least rigorous aspect of the work of UD, as they state this without proof.

Definition 5. A schedule S = (α, β) is pseudo-periodic if there exists an
increasing function ϕ : N → N such that:

(i) ϕ(0) = 0
(ii) ∀K ∈ N, i ∈ I. ∃t ∈ N. i ∈ α(t) ∧ ϕ(K) ≤ t < ϕ(K + 1)
(iii) ∀K, t ∈ N, i, j ∈ I. t ≥ ϕ(K + 1) =⇒ β(t, i, j) ≥ τi(K) ≥ ϕ(K)

where τi(K) is the earliest time after ϕ(K) that element i is updated.

The intuition behind ϕ is that by time ϕ(K + 1) every node is guaranteed
to be using data generated at least as recently as ϕ(K). Hence the interval
(ϕ(K), ϕ(K + 1)] is known as the kth pseudo-period.

We formalize the pseudo-periodic property in Agda as follows:

record IsPseudoperiodic {n : N} (S : Schedule n) : Set where
open Schedule S
field

ϕ : N → T

τ : N → Fin n → T

ϕ-increasing : ∀ K → K ≤ ϕ K
τ-active : ∀ K i → i ∈ α (τ K i)
τ-after-ϕ : ∀ K i → ϕ K ≤ τ K i
τ-expired : ∀ K t i j → τ K j ≤ B (ϕ (suc K) + t) i j

Note that this represents a simplification of UD’s definition. We worked
backwards from the proof of Theorem 1 and identified only those properties
required. This simplification may have to change if we extend our library to
include UD’s proof that the ACO condition is also necessary (in the case of
finite state spaces).

UD assert that for any schedule there exist an infinite number of possible
functions ϕ, but they do not provide any explicit constructions. This is one area
where we had initial concerns when planning our proof strategy in Agda.

We start by defining nextActive, which takes a time t and a node index i and
returns the first time after t for which that i is active.

nextActive’ : (t k : T) {i : Fin n} → i ∈ α (t + suc k) → Acc _<_ k → T

nextActive’ t zero {i} _ _ = suc t
nextActive’ t (suc k) {i} i∈α[t+1+K] (acc rs) with i ∈? α t
... | yes i∈α = t
... | no i/∈α rewrite +-suc t (suc k) = nextActive’ (suc t) k i∈α[t+1+K] _

nextActive : T → Fin n → T

nextActive t i with nonstarvation t i
... | (K , i∈α[t+1+K]) = nextActive’ t K i∈α[t+1+K] (<-wellFounded K)

We then define allActive, which returns the first time after t such that all nodes
have activated since t.

allActive : T → T

allActive t = max t (nextActive t)

We then need to define three auxiliary functions: pointExpiryij returns a time
after which i does not use the data generated by j at time t.

pointExpiryij : Fin n → Fin n → T → T

pointExpiryij i j t = proj1 (finite t i j)

expiryij returns a time after which i only uses data generated by j after time t.

expiryij : T → Fin n → Fin n → T

expiryij t i j = List.max t (applyUpTo (pointExpiryij i j) (suc t))

expiryi returns a time after which i only uses data generated after time t.

expiryi : T → Fin n → T

expiryi t i = max t (expiryij t i)

Using these we can define the function expiry that returns a time after which all
nodes only use data generated after time t.

expiry : T → T

expiry t = max t (expiryi t)

Finally, we construct ϕ as follows:

ϕ : N → T

ϕ zero = zero
ϕ (suc K) = suc (expiry (allActive (ϕ K)))

Therefore we find a time t such that all nodes have been activated after ϕ(K)
and then ϕ(K+1) is defined as the time after which all data used was generated
after t. The function τ (as defined in property (iii) of pseudo-periodic schedules)
is simply a special call to nextActive.

τ : N → Fin n → T

τ K i = nextActive (ϕ K) i

We now prove that ϕ and τ satisfy the properties required to be pseudo-periodic
as given in Definition 5. The property ϕ-increasing is relatively simple, given that
proofs that the various functions are increasing:

ϕ-increasing : ∀ K → K ≤ ϕ K
ϕ-increasing zero = z≤n
ϕ-increasing (suc K) = s≤s (begin

K ≤〈 ϕ-increasing K 〉
ϕ K ≤〈 allActive-increasing (ϕ K) 〉
allActive (ϕ K) ≤〈 expiry-increasing (allActive (ϕ K)) 〉
expiry (allActive (ϕ K)) �)

The second property says that τ is always active and it can be satisfied by using
properties of nextActive:

τ-active : ∀ K i → i ∈ α (τ K i)
τ-active K = nextActive-active (ϕ K)

The third property can be easily proved using the fact that nextActive is increas-
ing:

τ-after-ϕ : ∀ K i → ϕ K ≤ τ K i
τ-after-ϕ zero i = z≤n
τ-after-ϕ (suc K) i = nextActive-increasing (ϕ (suc K)) i

The final property states that at all points during a pseudo-period, no nodes use
information generated in a previous pseudo-period. This is the most complex of
the four properties to prove.

τ-expired : ∀ K t i j → τ K j ≤ B (ϕ (suc K) + t) i j
τ-expired K t i j = expiry-expired (begin
expiry (nextActive _ j) ≤〈 expiry-monotone (nextActive≤allActive _ j) 〉
expiry (allActive (ϕ K)) ≤〈 n≤1+n (expiry (allActive (ϕ K))) 〉
ϕ (suc K) ≤〈 m≤m+n (ϕ (suc K)) t 〉
ϕ (suc K) + t �) i j

As previously mentioned the construction of ϕ is not discussed in UD. Never-
theless, filling this gap required significant effort in our Agda development.

The proof of Theorem 1 requires an additional fact about the functions τi:
for each K, once all i have been updated after some time t, then x(t) ∈ D(K).

Lemma 1. ∀t, K ∈ N, i ∈ I. τi(K) ≤ t =⇒ xi(t) ∈ Di(K).

In UD Lemma 1 is proved by a fairly easy induction on K. However, in
Agda the construction, called τ-stability, turned out to be more difficult. Several
smaller lemmas were required, the biggest of which is that the asynchronous
iteration remains within D(0), the proof of which is called async[t]’∈D0.

async[t]’∈D0 : ∀ {t} (acct : Acc _<_ t) → async-Iter’ S x0 acct ∈ D 0
async[t]’∈D0 {zero} _ i = x0∈D0 i
async[t]’∈D0 {suc t} (acc rec) i with i ∈? α (suc t)
... | yes i∈α = D-decreasing 0 (F-monotonic 0 (ń j →

async[t]’∈D0 (rec (B (suc t) i j) (s≤s (causality t i j))) j)) i
... | no i/∈α = async[t]’∈D0 (rec t (s≤s ≤-refl)) i

τ-stability’ : ∀ {t K i} (acct : Acc _<_ t) → τ K i ≤ t →
async-Iter’ S x0 acct i ∈u D K i

τ-stability’ {_} {zero} {i} acct _ = async[t]’∈D0 acct i
τ-stability’ {zero} {suc K} {i} _ τ≤0 =

contradiction τ≤0 (<⇒� 0<τ[1+K])
τ-stability’ {suc t} {suc K} {i} (acc rec) τ≤1+t with i ∈? α (suc t)
... | yes _ = F-monotonic K (ń j → τ-stability’ _ (τ[1+K]-expired τ≤1+t)) i

... | no i/∈α with τ (suc K) i ?= suc t

... | no τ�≡1+t = τ-stability’ _ (<⇒≤pred (≤+�≡⇒< τ≤1+t τ�≡1+t))

... | yes τ≡1+t =
contradiction (subst (i ∈s_) (cong α τ≡1+t) (τ-active (suc K) i)) i/∈α

τ-stability : ∀ {t K i} → τ K i ≤ t → asyncIter S x0 t i ∈u D K i
τ-stability {t} = τ-stability’ (<-wellFounded t)

We now construct the final proof of convergence. To do this we must construct
a time after which the result of the asynchronous iteration is always equal to
the fixed point. UD prove that ϕ(T + 1), where T is from the ACO, is the
convergence time. This is because each pseudo-period, every node is updated at
least once and a total of T updates must occur before convergence. In the Agda,
we first extract T and ξ from D-Finish. We then prove Theorem 1 as follows.

T : T

T = proj1 D-finish

ξ : S
ξ = proj1 (proj2 D-finish)

tc : T

tc = F (suc T)

async[tc]∈D[T] : ∀ t → asyncIter S x0 (tc + t) ∈ D T
async[tc]∈D[T] t j = τ-stability (begin
τ T j ≤〈 τ-expired T 0 j j 〉
B (tc + 0) j j ≡〈 cong (ń v → B v j j) (+-identityr tc) 〉
B tc j j ≤〈 B-decreasing j j 1≤tc 〉
tc ≤〈 m≤m+n tc t 〉
tc + t �)
where open ≤-Reasoning

async-converge : ∀ K → asyncIter S x0 (tc + K) ≈ ξ
async-converge K = D[T]≈{ξ} (async[tc]∈D[T] K)

4 The Library

UD show that being an ACO is a sufficient (and sometimes a necessary) condi-
tion for convergence. However in practice, constructing the sets D(K) can still
be a non-trivial exercise. Therefore, an extensive array of sufficient (but often
not necessary) conditions have been constructed that in practice can be simpler
and more intuitive to apply. These conditions are nearly always a reduction back
to ACOs.

In this section we discuss three different proposed sufficient conditions. The
first two are from UD and the third is a modified version of a new sufficient
condition found in a recent paper by Gurney [11] (which was essential for the
results described in Daggitt et al. [5]). During the formalization process, we
discovered counterexamples to the two conditions from UD.

4.1 Synchronous Iteration Conditions

The first set of sufficient conditions makes use of the synchronous iteration of the
algorithm, which UD refer to as y(t), as opposed to the asynchronous iteration
x(t). The conditions involve the existence of partial orderings, �i, over each Si,
which are lifted to the partial order � over S in the usual point-wise manner.
UD then make the following claim (Proposition 3 in UD):

Claim 1. An operator F has a fixed-point ξ to which every asynchronous iter-
ation converges for every starting state y(0) ∈ D(0) if:

(i) ∀a ∈ D(0). F(a) ∈ D(0)
(ii) ∀a , b ∈ D(0). a � b =⇒ F(a) � F(b)
(iii) ∀K ∈ N. y(K + 1) � y(K)
(iv) The sequence {y(K)} converges.

They attempt to prove this by first showing a reduction from these conditions
to an ACO and then using Theorem 1 to obtain the required result.

However this claim is not true. While the asynchronous iteration does con-
verge from every starting state in D(0), it does not necessarily converges to the
same fixed point. The flaw in the original proof is that UD tacitly assume that
the set D(0) for the ACO they construct is the same as the original D(0) speci-
fied in the conditions above. However the only elements that are provably in the
ACO’s D(0) is the set {y(t) | t ∈ N}. We now present a counter-example to the
claim.

Consider the degenerate asynchronous environment that contains only a sin-
gle node (i.e. I = {0}) and let F be the identity function (i.e. F(a) = a). Let
D(0) = {x, y} where the only relationships in the partial order are x � x and
y � y. Clearly (i), (ii), (iii) and (iv) all trivially hold as F is the identity function.
However x and y are both fixed points, and which fixed point is reached depends
on whether the iteration starts at x or y. Hence Claim 1 cannot be true.

We can strengthen the conditions by changing requirement (iv) to “There
exists a ξ such that for all y(0) the sequence {y(K)} converges to ξ”. The
library formalises these conditions in Agda as:

record SynchronousConditions p o : Set (lsuc (a � � � p � o)) where

field
D0 : Pred Si p
D0-cong : ∀ {x y} → x ∈ D0 → x ≈ y → y ∈ D0

D0-closed : ∀ {x} → x ∈ D0 → F x ∈ D0

_≤_ : Rel Si o
≤-isPartialOrder : IsIndexedPartialOrder S _≈_ _≤_

F-monotone : ∀ {x y} → x ∈ D0 → y ∈ D0 → x ≤ y → F x ≤ F y
F-cong : ∀ {x y} → x ≈ y → F x ≈ F y
iter-decreasing : ∀ {x} → x ∈ D0 → ∀ K → syncIter x (suc K) ≤ syncIter x K

ξ : S
ξ-fixed : F ξ ≈ ξ
iter-converge : ∀ {x} → x ∈ D0 → ∃ ń T → syncIter x T ≈ ξ

The reduction of these conditions to an ACO runs as follows. The sequence
of sets D required by the definition of an ACO are defined as follows:

D(K) = {x | ξ � x � y(K) ∧ x ∈ D0}

which is directly translated in Agda as:

D : N → ∀ i → Mi i → Set p
D K i = (ń x → (ξ i � x) × (x � sync-iter x0 K i)) ∩ D0 i

The field D-decreasing can be proven using iter-decreasing and D-finish is a
consequence of iter-converge and ξ-fixed. F-monotonic is the same for both ACO
and SynchronousConditions.

Routing Example. Classical routing theory [2] assumes that distributivity
holds:

∀e ∈ E : x, y ∈ S : e(x ⊕ y) = e(x) ⊕ e(y) (2)

and under this assumption one can prove that every entry of every routing table
improves monotonically with each iteration when the protocol starts from the
initial state I. Therefore for classical routing problems such as shortest-paths, it
is relatively easy to construct an instance of SynchronousConditions.

4.2 Finite Conditions

Another set of sufficient conditions proposed by UD are applicable when the
initial set D(0) is finite. Like Proposition 1, it requires that F is monotonic and
D(0) be closed over F. Instead of reasoning about the synchronous iteration of
the operator, it adds an additional requirement that F is non-expansive over
D(0).

Claim 2. An operator F has a fixed-point ξ to which every asynchronous iter-
ation converges for every starting state y(0) ∈ D(0) if:

(i) D(0) is finite
(ii) ∀a ∈ D(0). F (a) ∈ D(0)
(iii) ∀a ∈ D(0). F (a) � a
(iv) ∀a , b ∈ D(0). a � b =⇒ F (a) � F (b)

UD’s attempted proof for Claim 2 is a reduction to the conditions for Claim 1.
Like Claim1, the conditions therefore guarantee convergence but not to a unique
solution. Similarly the counterexample for Claim 1 is also a counterexample for
Claim 2.

Unlike Claim 1, we do not have a proposed strengthening of Claim 2 which
would guarantee the uniqueness of the fixed point. The reason is that the finite-
ness condition, while guaranteeing the existence of a fixed point when combined
with the other conditions, does not help to prove uniqueness. Instead much
stronger conditions would be required, for example the assumption of the exis-
tence of a metric space over the computation as discussed in the next subsection.
Any such stronger conditions tend to make finiteness superfluous.

4.3 Ultrametrics

The notion of convergence has an intuitive interpretation in metric spaces. In
such spaces, convergence is equivalent to every application of the operator F
moving you closer (in discrete steps) to the fixed point ξ.

There already exist results of this type. For instance El Tarazi [8] shows
that if there is a normed linear space over each the values at each node i, then
convergence occurs if there exists a fixed point x∗ and a γ ∈ (0, 1] such that:

||F(x) − x∗|| � γ||x − x∗||
However in many ways this is a very strong sufficient condition as the existence
of a norm requires the existence of an additive operator on the space. For many
processes, including our example of network routing, this may not be true.

Instead there is a more general result by Gurney [11] based on ultrametrics.
An ultrametric [19] is a metric where the standard triangle inequality has been
replaced by the strong triangle inequality. As far as we are aware, this result
seems to have appeared only in [11]. The work proves that the ultrametric con-
ditions not only imply the existence of an ACO but are actually equivalent to the
existence of an ACO and therefore equivalent to saying the process converges.
As with the theorems of UD we are primarily concerned with the usability of
the theorems and therefore only prove the forwards direction.

Definition 6. An ultrametric space (S, Γ, d) is a set S, a totally ordered set Γ
with a least element 0, and a function d : S → S → Γ such that:

M1 : d(x, y) = 0 ⇔ x = y
M2 : d(x, y) = d(y, x)
M3 : d(x, z) � max(d(x, y), d(y, z))

Definition 7. A function f : S → S is strictly contracting on orbits in an
ultrametric space (S, Γ, d) if:

x �= f(x) =⇒ d(x, f(x)) > d(f(x), f(f(x)))

i.e. the distance between iterations strictly decreases.

Definition 8. An operator f : S → S is strictly contracting on a fixed point
x∗ in an ultrametric space (S, Γ, d) if:

x �= x∗ =⇒ d(x∗, x) > d(x∗, f(x))

Theorem 2 (Gurney [11]). If there exists (Si, Γ, di), and we take S =
∏

i Si

and d(x,y) = maxi di(xi,yi) then F is an ACO if:

1. Γ is finite
2. F is strictly contracting on orbits over (S, Γ, d)
3. F is strictly contracting on a fixed point over (S, Γ, d)
4. S is non-empty

These conditions are constructed in Agda as:

record UltrametricConditions : Set (a � 
) where
field
di : ∀ {i} → Si i → Si i → N

d : S → S → N

d x y = max 0 (ń i → di (x i) (y i))

field
di-isUltrametric : ∀ {i} → IsUltrametric (Si i) di
F-strContrOnOrbits : F StrContrOnOrbitsOver d
F-strContrOnFP : F StrContrOnFixedPointOver d
d-bounded : Bounded d

element : S

_ ?=_ : Decidable _≈_
F-cong : F Preserves _≈_ −→ _≈_

Note that in our formalisation we currently assume Γ = Fin n for some n in
order to simplify the theory. We plan to generalize this at some point.

Our Agda proof is very similar to the original proof by Gurney [11]. One of
the key differences is that Gurney assumes that F is contracting where as we
assume that F is strictly contracting on a fixed point. This is because in our
use-case it is not possible to construct a contracting metric. The relationship
between the two properties is not entirely clear, but the resulting proofs are
very similar.

Routing Example. The Border Gateway Protocol [18] is used by all Inter-
net Service Providers (ISPs) to maintain connectivity in the global internet.
As explained in [5], distributivity (Eq. 2) cannot be guaranteed in this setting
primarily because of the competing interests of service providers and the very
expressive policy languages needed to implement these interests in routing.

Consequently, a great deal of research has been directed at finding sufficient
conditions that guarantee convergence for policy-rich protocols such as BGP (see
for example [10,20]). One reasonable condition is that the algebra be strictly
increasing :

∀e ∈ E : x ∈ S : x = x ⊕ e(x) �= e(x) (3)

This says that a route x must be strictly more preferred than any extension e(x).
However, now individual routing table entries are no longer guaranteed to

improve monotonically, and so there is no natural ordering on the state space.
Assuming Eq. 3, [5] show how to construct suitable ultrametrics di over the rout-
ing tables in such a way that they fulfill the properties required by Theorem 2.
It is based on the observation that the worst routing table entry in the state will
always improve after each iteration.

5 Conclusion

In this paper we have taken the mathematically rigorous yet informal proof
of Üresin and Dubois’ theory regarding the convergence of asynchronous itera-
tions [21] and formalized it constructively in Agda. After explicitly constructing
the previously unspecified pseudo-periodic sequences and mildly weakening some
assumptions, we have succeeded in formalizing the core theorem of the paper.
However some of the auxiliary propositions proposed in the original paper turned
out to be false. In our opinion this alone justifies the formalization process.

Furthermore, we have described our library of proofs and sufficient conditions
for asynchronous convergence, including a recent, new ultrametric condition. We
hope that the library of sufficient conditions will be a valuable resource for those
wanting to formally verify the convergence of a wide range of asynchronous
iterations. The library is available on Github [1].

We are primarily interested in proving convergence and therefore we have
thus far only formalized the sufficient conditions from Üresin and Dubois and
not their proof that the ACO condition is also necessary in the case of finite state
spaces. This would be an interesting extension to our development. In addition
it would be interesting to see if other related work such as [15,22,23], using
different models, could be integrated into our formalization.

